

# DustemWrap V1.0 User's Guide

J.Ph. Bernard, D. Paradis, N. Flagey

June 8, 2011

`DustemWrap` is an IDL code which allows to use the `Dustem` Fortran code from within IDL. It allows to iteratively call the Fortran and to modify the input parameters from inside IDL. This is particularly useful for the following applications:

1. Compute color corrections in photometric bands for any `Dustem` emission spectrum
2. Iteratively fit a given astronomical SED (Spectral Energy Distribution) with `Dustem`
3. Compute tables of pre-computed `Dustem` outputs for fast comparisons with astronomical data

The applications of `DustemWrap` are not limited to the list above. As `DustemWrap` allows to run `Dustem` from IDL, any application requiring multiple call to `Dustem` can be easily implemented. Note also that `DustemWrap` implements transmission curves and color correction rules for a large set (more than 10) of past and present astronomical missions (see Sec. ??). Therefore, `DustemWrap` can also be used to easily construct observed SEDs from any model prediction and not just `Dustem` predictions. Finally, since the `DustemWrap` fitting tool is based on `mpfit`, the `DustemWrap` environment can also be used to implement SED fits which are not based on the `Dustem` prediction, but on other IDL-based models.

The `Dustem` Fortran code itself and the associated documentation are available at :  
<http://www.ias.u-psud.fr/DUSTEM>

The `DustemWrap` IDL code and the associated documentation are available at :  
[http://www.cesr.fr/DUSTEM\\_WRAP](http://www.cesr.fr/DUSTEM_WRAP)

We describe how to install `DustemWrap` in Sec.1. Sec.7 gives a short description of all routines used. Sec.6 shows a few examples of use.

## 1 Installation

1. Download the `Dustem` Fortran from the following web site:  
(<http://www.ias.u-psud.fr/DUSTEM>)  
Note that this version of `DustemWrap` has been tested only for the V3.8 of `Dustem` .
2. In order to run the wrapper, you will have to edit the file `DM_constants.f90` of the Fortran `Dustem` release, to set the `data_path` variable to point to a region of your disk where you want the wrapper to store its data.  
Warning: You should use an initially **\*empty\*** directory for this, since everything there will be erased by the wrapper.  
For instance:  

```
CHARACTER (len=100) :: data_path='/tmp/dustem/'
```

  
You then should recompile the `Dustem` code, following the instructions given in the `Dustem` documentation.
3. Compile the `Dustem` Fortran following the instructions in the `Dustem` Fortran package.

4. After downloading the DustemWrap tar ball from the DustemWrap website ([http://www.cesr.fr/DUSTEM\\_WRAP](http://www.cesr.fr/DUSTEM_WRAP))  
unpack the archive:  
`tar xvjf dustemwrap.tar.gz`  
this will generate a directory named DUSTEM\_WRAP
5. move the DUSTEM\_WRAP directory where you want it to stay on your disk, e.g.  
`mv DUSTEM_WRAP $HOME/Soft_Libraries/DUSTEM_WRAP/`
6. Include the following lines in your `idl_startup` file, making sure to replace the variable values by those matching your installation (values below are given for the example install of this document):

```

;===DUSTEM_WRAP release V1.0
defsysv,'!dustem_soft_dir','$HOME/Soft_Libraries/DUSTEM/'
defsysv,'!dustem_wrap_soft_dir','$HOME/Soft_Libraries/DUSTEM_WRAP/'
defsysv,'!dustem_dat','/tmp/dustem/'
defsysv,'!dustem_res','/tmp/dustem/'
defsysv,'!dustem_which','WEB3p8'
defsysv,'!dustem_f90_exec','$HOME/Soft_Libraries/DUSTEM/src/dustem3.8_web/src/dustem'

```

Note that the / character is needed at the end of paths.

Note that `!dustem_f90_exec` must point to the Dustem Fortran executable as produced in step 2.

7. Include the following line in your `idl_startup` file, which will add the DustemWrap code into your `idl` path  
`!path=!path+' '+expand_path(''+!dustem_wrap_soft_dir+ '/src/idl')`
8. Optional: If in addition, you do not wish to use the external libraries which DustemWrap needs (see Sec. 8), also include the following line in your `idl_startup` file, which will add the `idl_misc` directory to your path :  
`!path=!path+' '+expand_path(''+!dustem_wrap_soft_dir+ '/src/idl_misc')`
9. Now under IDL, you should be able to run the `dustem_init` initialisation of the wrapper  
`IDL> dustem_init,/wrap,/plot_it`

There are a few options to the above commands, which allow to work with the different dust properties known by Dustem . See for instance:

```
IDL> dustem_init,/help
```

## 2 How does DustemWrap work ?

Any use of DustemWrap require to initialize a few system variables used by the wrapper. This operation is done using the routine `dustem_init.pro`. A full description of the system variable used is given in Sec. 9.

DustemWrap has 3 main functionalities, which allow to run Dustem from within IDL:

- Read the Fortan input files. Those contain parameters for the Fortran, such as dust composition, Interstellar radiation field, dust optical cross sections, etc ... The are described in details in the Dustem documentation.
- Store the above inputs into an IDL variable. This variable is an IDL system variable (called `!dustem_params`) which is available from any IDL routine. This allows the user or the wrapper to modify the inputs parameters from within IDL.

- Write the input parameters into a set of files which can be read by the `Dustem` Fortran code. These modified files are written in directory `DUSTEM_DAT`
- Launch `Dustem` on the above files. This is done through the `dustem_run.pro` function.
- Read the `Dustem` outputs and store the results into an IDL structure. This structure is a regular IDL variable.

An example of how to perform the above operations is given in `dustem_run_readme.pro` provided with `DustemWrap` .

The output structure (st) of `dustem_run.pro` has the following form:

st.dustem:

WAV : emission wavelengths (microns)

EM\_GRAIN\_1 : emission of grain type # 1 ( $4\pi\nu I_{\nu}$  in  $erg/s/cm^2/H$ )

EM\_GRAIN\_2 : emission of grain type # 2 ( $4\pi\nu I_{\nu}$  in  $erg/s/cm^2/H$ )

etc ...

EM\_TOT : Total emission ( $4\pi\nu I_{\nu}$  in  $erg/s/cm^2/H$ )

st.ext:

WAV : extinction wavelengths (in microns)

ABS\_GRAIN : Absorption of each grain type per dust mass (in  $cm^2/g$ )

SCA\_GRAIN : Scattering of each grain type per dust mass (in  $cm^2/g$ )

In addition to the above, `DustemWrap` also provides tools to iteratively fit SEDs using `Dustem` . The fit is performed following the steps below:

- Read an astronomical SED and store it into a dedicated IDL system variable (called `!dustem_data`). This is performed using `dustem_set_data.pro`. SEDs can be filled "manually" from within an IDL piece of code, or, more readily, read from a text file with extension `.xcat`. An example of such an input SED file is given in `Gal_composite_spectrum.xcat` provided with `DustemWrap` [At the moment in a `Dustem` directory: to be moved]. Note that SEDs must include the name of the filter used, the central wavelength [optional], the brightness or flux, the  $1 - \sigma$  uncertainty on the brightness or flux and optionally, the flux convention used. A detailed description of an input SED is given in Sec 3
- Declare variables which values will be changed compared to the "default" values given in the `Dustem` Fortran input files, and specify their values.
- Declare which variables of the model will be varied during minimization, specifying their possible range of variations and initial values.
- Perform minimization. This is made using the `dustem_mpfite_sed_routine.pro`, which uses the `mpfitfun.pro` function, which is available from the `cmtotal` IDL library. During minimization, the output spectrum of `Dustem` are color-corrected according to the flux convention used by each instrument and using the filter transmission of the filters. Derivatives of the model with respect to the variable input parameters are computed numerically.
- Recover the fit results. This includes best fit parameter values and associated errors,  $\chi^2$  and reduced  $\chi^2$  for the fit.
- Optionally save the fit results for later used.

An example of how to perform the above operations is given in `dustem_fit_sed_readme.pro` provided with `DustemWrap` . The output of this example is shown in Fig. 1.




Figure 1: Example of SED fitting using `DustemWrap`. The red points show the SED data, composed of spectral data (dots) and photometric data (squares). The yellow (spectral) and blue (photometric) dots show the best fit SED. The dashed lines show the spectra for the various dust components used in the fit. The values printed on the plot give the best fit values of the free parameters and their uncertainties. The  $\chi^2$  and reduced  $\chi^2$  are also shown.

### 3 SED description

SED information to be used by `DustemWrap` are provided through the `dustem_set_data.pro` routine (see example in Sec. 6.3). The easiest way to construct the idl structure containing SED information to be passed to `dustem_set_data.pro` is to read it from a text file (called `.xcat` file) such as shown bellow, using the `read_xcat.pro` routine (see example in Sec. 6.3). An example of such an SED file is available in the `DustemWrap` package (`Gal_composite_spectrum.xcat`).

Note that such SED files can be constructed by generating an empty structure (using `dustem_make_sed_st.pro`), filling it with the appropriate data and saving it into a `.xcat` file using the `write_xcat.pro` routine.

```

\Written by write_xcat.pro on Thu Mar 31 16:19:48 2011
\Dust High Galactic Latitude SED
\for NH=10^20 H/cm2
|INSTRU |FILTER |WAVE |SPEC |ERROR |UNIT |
|char |char |real |real |real |char |
|NULL |NULL |NULL |NULL |NULL |NULL |
FIRAS SPECTRUM 1200.00 0.0130615 0.00313187 MJy/sr
FIRAS SPECTRUM 1102.60 0.0191160 0.00629346 MJy/sr
FIRAS SPECTRUM 1050.10 0.0169294 0.0135306 MJy/sr
....
FIRAS SPECTRUM 128.210 0.966572 0.368282 MJy/sr
FIRAS SPECTRUM 127.469 0.940774 0.620879 MJy/sr
FIRAS SPECTRUM 126.736 1.09370 0.874983 MJy/sr
DIRBE DIRBE7 60.0000 0.0989657 0.00213576 MJy/sr
DIRBE DIRBE8 100.000 0.554668 0.0633852 MJy/sr
DIRBE DIRBE9 140.000 1.05070 0.0876703 MJy/sr
DIRBE DIRBE10 240.000 0.825213 0.0712451 MJy/sr
WMAP WMAP1 13043.5 0.000254173 2.62925e-05 MJy/sr
\WMAP WMAP2 9090.91 0.000171453 1.56830e-05 MJy/sr

```

\WMAP	WMAP3	7317.07	0.000136109	9.82698e-06	MJy/sr
\WMAP	WMAP4	4918.03	0.000140426	8.04632e-06	MJy/sr
\WMAP	WMAP5	3191.49	0.000442967	1.61845e-05	MJy/sr
ISO	SPECTRUM	5.07900	0.000918936	0.00009	MJy/sr
ISO	SPECTRUM	5.14100	0.000813131	0.00008	MJy/sr
....					
ISO	SPECTRUM	16.4200	0.0139343	0.00100	MJy/sr
ISO	SPECTRUM	16.5200	0.0143230	0.00100	MJy/sr
DIRBE	DIRBE3	3.53000	0.00114137	0.000228273	MJy/sr
DIRBE	DIRBE4	4.90000	0.00181300	0.000362600	MJy/sr
DIRBE	DIRBE5	12.0000	0.0286400	0.00572800	MJy/sr
DIRBE	DIRBE6	25.0000	0.0297500	0.00595000	MJy/sr
\AROME	AROME	3.30000	0.00635506	0.00288866	MJy/sr

The SED information includes the following fields:

- INSTRU: Instrument name. This information is not used by `DustemWrap`
- FILTER: Filter name. Those are described in Sec. 4. The "SPECTRUM" value indicates spectral data for which no color correction is required.
- WAVE: This is the reference wavelength for filter measurements and the wavelength of the data for spectral data. Note that for filter data, the value entered here will be replaced by `DustemWrap` during SED fitting by the reference value corresponding to the corresponding filter. The reference wavelength used by `DustemWrap` for a given filter is the one returned by `dustem_filter2wav.pro`
- SPEC: This is the brightness (specific intensity) or flux value. Note that `DustemWrap` computes its outputs for a gas column density of  $N_H = 10^{20} H/cm^2$ . Therefore, fitting results obtained for a SED computed for this gas column density will require no scaling before interpreting the results.
- ERROR: This is the  $1 - \sigma$  uncertainty associated with SPEC in the same units as SPEC.
- UNIT: Unit of SPEC and ERROR. This information is not used by `DustemWrap` .

## 4 Available filters

Filters available in `DustemWrap` are summarized in Tab. 1. Each filter is attributed a unique name which starts with the instrument name, followed by a number starting from 1 and increasing with the reference wavelength of the filter. This allows to refer to a given filter through this unique name in SED data, without specifying the associated reference wavelength or frequency or the flux convention. This data is stored in a table called `instrument_description.xcat`. `DustemWrap` includes routines allowing to derive other parameters (e.g. Instrument name, central frequency, central wavelength, flux convention) from a given filter name.

IRAS transmissions are assumed to be the same transmission as for the DIRBE corresponding filters (TBC).

ISOPHOT filter transmissions were obtained from R. Laureijs and ?

Archeops filter transmissions were obtained from X. Désert

WMAP filter transmissions were originally provided by G. Hinshaw. They have since been made available on the Lambda web site.

## 5 Color Corrections

Photometric instrument measure astronomical spectra within a finite bandwidth in frequency. The spectral response within the bandwidth is set by the optical elements of the instrument,

Table 1: Filters currently implemented in DustemWrap and associated flux conventions.

Instrument	Filter	Ref. Wavelengths ( $\mu\text{m}$ )	Flux Convention
IRAC	IRAC1	3.550	IRAC
IRAC	IRAC2	4.493	IRAC
IRAC	IRAC3	5.731	IRAC
IRAC	IRAC4	7.872	IRAC
IRAS	IRAS1	12.000	nuInu=cste
IRAS	IRAS2	25.000	nuInu=cste
IRAS	IRAS3	60.000	nuInu=cste
IRAS	IRAS4	100.000	nuInu=cste
AKARI	AKARI1	65.000	nuInu=cste
AKARI	AKARI2	90.000	nuInu=cste
AKARI	AKARI3	160.000	nuInu=cste
AKARI	AKARI4	140.000	nuInu=cste
MIPS	MIPS1	23.680	MIPS
MIPS	MIPS2	71.420	MIPS
MIPS	MIPS3	155.900	MIPS
WMAP	WMAP1	3200.000	CMB
WMAP	WMAP2	4900.000	CMB
WMAP	WMAP3	7300.000	CMB
WMAP	WMAP4	9100.000	CMB
WMAP	WMAP5	13000.000	CMB
MSX	MSX1	8.280	nuInu=cste
MSX	MSX2	12.130	nuInu=cste
MSX	MSX3	14.650	nuInu=cste
MSX	MSX4	21.340	nuInu=cste
DIRBE	DIRBE1	1.250	nuInu=cste
DIRBE	DIRBE2	2.200	nuInu=cste
DIRBE	DIRBE3	3.500	nuInu=cste
DIRBE	DIRBE4	4.900	nuInu=cste
DIRBE	DIRBE5	12.000	nuInu=cste
DIRBE	DIRBE6	25.000	nuInu=cste
DIRBE	DIRBE7	60.000	nuInu=cste
DIRBE	DIRBE8	100.000	nuInu=cste
DIRBE	DIRBE9	140.000	nuInu=cste
DIRBE	DIRBE10	240.000	nuInu=cste
ARCHEOPS	ARCHEOPS1	550.078	CMB
ARCHEOPS	ARCHEOPS2	849.270	CMB
ARCHEOPS	ARCHEOPS3	1381.532	CMB
ARCHEOPS	ARCHEOPS4	2096.451	CMB
HFI	HFI1	349.816	nuInu=cste
HFI	HFI2	550.078	nuInu=cste
HFI	HFI3	849.270	nuInu=cste
HFI	HFI4	1381.532	nuInu=cste
HFI	HFI5	2096.451	nuInu=cste
HFI	HFI6	2997.925	nuInu=cste
LFI	LFI1	4285.710	nuInu=cste
LFI	LFI2	6818.180	nuInu=cste
LFI	LFI3	10000.000	nuInu=cste
SPM	SPM1	200.000	nuInu=cste
SPM	SPM2	260.000	nuInu=cste
SPM	SPM3	360.000	nuInu=cste
SPM	SPM4	580.000	nuInu=cste
ISOPHOT	ISOPHOTP1	3.300	nuInu=cste
ISOPHOT	ISOPHOTP2	3.600	nuInu=cste
ISOPHOT	ISOPHOTP3	4.800	nuInu=cste
ISOPHOT	ISOPHOTP4	7.300	nuInu=cste
ISOPHOT	ISOPHOTP5	7.700	nuInu=cste
ISOPHOT	ISOPHOTP6	10.000	nuInu=cste
ISOPHOT	ISOPHOTP7	11.300	nuInu=cste
ISOPHOT	ISOPHOTP8	12.000	nuInu=cste
ISOPHOT	ISOPHOTP9	12.800	nuInu=cste
ISOPHOT	ISOPHOTP10	15.000	nuInu=cste
ISOPHOT	ISOPHOTP11	20.000	nuInu=cste
ISOPHOT	ISOPHOTP12	25.000	nuInu=cste
ISOPHOT	ISOPHOTP13	60.000	nuInu=cste
ISOPHOT	ISOPHOTP14	100.000	nuInu=cste
ISOPHOT	ISOPHOTC1	65.000	nuInu=cste
ISOPHOT	ISOPHOTC2	60.000	nuInu=cste
ISOPHOT	ISOPHOTC3	80.000	nuInu=cste
ISOPHOT	ISOPHOTC4	90.000	nuInu=cste
ISOPHOT	ISOPHOTC5	100.000	nuInu=cste
ISOPHOT	ISOPHOTC6	105.000	nuInu=cste
ISOPHOT	ISOPHOTC7	120.000	nuInu=cste
ISOPHOT	ISOPHOTC8	150.000	nuInu=cste
ISOPHOT	ISOPHOTC9	170.000	nuInu=cste
ISOPHOT	ISOPHOTC10	180.000	nuInu=cste
ISOPHOT	ISOPHOTC11	200.000	nuInu=cste

such as the filters. The measurements in each filter (or bandwidth) are referred to a reference wavelength  $\lambda_0$ . The in-band measurements correspond to the intensity at  $\lambda_0$  of a spectrum with a specific spectral shape (called the flux convention) giving the same power as was measured by the instrument. Different missions have used different flux conventions. The most popular flux convention is the one originally adopted by the IRAS experiment and corresponds to "nu\*I<sub>nu</sub>=cste". In that case, the spectral shape of the fiducial spectrum is  $I_\nu \propto \nu^{-1}$ .

Color correction factors ( $K$ ) are defined as the factor by which a given spectrum at wavelength  $\lambda_0$  ( $I_\nu(\nu_0)$ ) must be multiplied to obtain the measurement a given photometric band in the considered flux convention ( $\tilde{I}_\nu^0$ ):

$$\tilde{I}_\nu^0 = K \times I_\nu(\nu_0), \quad (1)$$

This parameter depends on the spectral shape of both the instrument transmission and the input spectrum.

Equating the in-band power of the reference spectrum and the actual spectrum for the "nu\*I<sub>nu</sub>=cste" flux convention ( $I_\nu = \tilde{I}_\nu^0 \times (\nu/\nu_0)^{-1}$ ) leads to :

$$\int_0^\infty \tilde{I}_\nu^0 \left(\frac{\nu}{\nu_0}\right)^{-1} T(\nu) d\nu = \int_0^\infty T(\nu) I_\nu d\nu, \quad (2)$$

where  $T(\nu)$  is the filter transmission.

The color correction factor for the "nu\*I<sub>nu</sub>=cste" flux convention can therefore be computed as:

$$K = \frac{1}{\nu_0 I_\nu(\nu_0)} \times \frac{\int_0^\infty T(\nu) I_\nu d\nu}{\int_0^\infty T(\nu) \nu^{-1} d\nu}, \quad (3)$$

or equivalently for a wavelength integration:

$$K = \frac{\lambda_0}{I_\nu(\lambda_0)} \times \frac{\int_0^\infty T I_\nu \lambda^{-2} d\lambda}{\int_0^\infty T \lambda^{-1} d\lambda}, \quad (4)$$

In `DustemWrap` , the color correction are computed using the routine `dustem_cc.pro` (see Sec. 7), which computes the SED value ( $\tilde{I}_\nu^0$ ) for a set of instrument filters, given an input spectrum specified by its wavelengths and specific intensity values. The routine optionally returns the color correction factors. The flux convention used for the color correction can be specified but defaults to the default convention for each filter listed in Tab. 1.

Available flux conventions are listed below:

- "nu\*I<sub>nu</sub>=cste": The most widely used. Assumes a reference spectrum  $I_\nu = \tilde{I}_\nu^0 \times (\nu/\nu_0)^{-1}$
- "IRAC": The flux convention used by the IRAC instrument on the Spitzer satellite. It is similar to the "nu\*I<sub>nu</sub>=cste" convention, but the spectral response used is in electrons/photons instead of ergs/photons, leading to a slightly different formula (see ? for details).
- "MIPS": The flux convention used by the MIPS instrument on the Spitzer satellite. The reference spectrum is a BlackBody at  $T = 10^4 K$  (see ? for details).
- "CMB": The flux convention for measurements originally provided in CMB temperature units ( $K_{CMB}$ ), which have been simply transformed to specific intensity using  $I_\nu = T_{CMB} \times (dB_\nu/dT)(T_{CMB})$ . The reference spectrum in that case is assumed to have  $I_\nu \propto (dB_\nu/dT)(T_{CMB})$ .

## 6 Examples of use

### 6.1 Running Dustem from IDL

The following code shows an example of how to run `Dustem` using `DustemWrap` . Those lines are taken from the routine `dustem_run_readme.pro` provided with the `DustemWrap` release.

```

;== Initialize DustemWrap
dustem_init,mode='COMPIEGNE_ETAL2010'

;== set DustemWrap behavior
!dustem_verbose=1
!dustem_show_plot=1

;== The following structure contains the inputs to the model
st_model=dustem_read_all(!dustem_soft_dir)

;== The following line writes inputs to the Fortran
dustem_write_all,st_model,!dustem_dat
;== This runs the Fortran. The output structure contains the results
st=dustem_run()

;== The following just plots the resulting emission SED
loadct,13
window,0
dustem_plot_nuinu_em,st,yr=[1e-11,6.e-7],/ysty,xr=[1,5e3],/xsty
;== The following just plots the resulting extinction
window,1
dustem_plot_extinction,st,st_model,yr=[1.e-6,1.e0],/ysty,xr=[1.,400],/xsty

```

## 6.2 Color Correction

The following code shows an example of how to compute color corrections with Dustem. The following code lines show how to reproduce the MIPS color correction factors for power law spectra. The corresponding results can be compared with the table given in ?. They are taken from the code `dustem_cc_mips_vs_handbook.pro` which also shows an example involving black-body spectra, and is included in the `DustemWrap` release.

```

;== Initialize DustemWrap
dustem_init

;== Define filters
filters=['MIPS1','MIPS2','MIPS3']
Nfilt=n_elements(filters)
;== Define Wavelengths for spectrum definition
Nwav=1000 & wavmin=10. & wavmax=1000.
wavs=findgen(Nwav)/(Nwav-1)*(wavmax-wavmin)+wavmin

;== Compute color corrections for power laws of frequency
pows=[3.,-2.,-1.,0.,1,2,3]
Nv=n_elements(pows)
K=fltarr(Nv,Nfilt)
frmt1='(15F9.3)'
frmt2='(15I9)'
FOR i=0L,Nv-1 DO BEGIN
  spec=wavs^(-1.*pows(i))
  sed=dustem_cc(wavs,spec,filters,cc=cc)
  K(i,*)=cc
ENDFOR

;== print results
print,pows,format=frmt2
FOR i=0L,Nfilt-1 DO print,K(*,i),format=frmt1

```



### 6.3 Fitting SED

The routine `dustem_fit_sed_readme.pro` contains an example of fitting an SED with `DustemWrap`. Under IDL, you should be able to test it using:

```
IDL> dustem_fit_sed_readme
```

There are a few options to the above command, which allow to work with the different dust properties known by `dustem`. See for instance

```
IDL> dustem_fit_sed_readme,/help
```

The following lines are taken from the above code and illustrate how to fit a SED with `DustemWrap`.

```
;;= Initialize DustemWrap
dustem_init

;;= set DustemWrap behavior
!dustem_verbose=1
!dustem_show_plot=1

;;= Read an SED
dir=!dustem_wrap_soft_dir+'/Data/SEDs/'
file=dir+'Gal_composite_spectrum.xcat'
spec=read_xcat(file,/silent)
;=== SET THE OBSERVATION STRUCTURE
dustem_set_data,spec

;;= define which model parameters to vary
;=== pd is the parameter definition
pd=[$
  '(!dustem_params).G0', $ ;G0 radiation field intensity
  '(!dustem_params).grains(1).mdust_o_mh', $ ;PAH1 mass fraction
  '(!dustem_params).grains(0).mdust_o_mh', $ ;PAH0 mass fraction
  '(!dustem_params).grains(2).mdust_o_mh', $ ;amCBEx mass fraction
  '(!dustem_params).grains(3).mdust_o_mh', $ ;amCBEx mass fraction
  '(!dustem_params).grains(4).mdust_o_mh', $ ;aSil mass fraction
  'dustem_create_continuum_2'] ;Intensity of NIR continuum
Npar=n_elements(pd)
;=== iv gives the corresponding initial values
iv = [1.0, 7.8e-4, 7.8e-4, 1.65e-4, 1.45e-3, 7.8e-3, 0.001]
;=== The following indicate if the parameters are upper or lower limited
ulimed=replicate(0,Npar)
llimed=replicate(1,Npar)
;=== The following gives corresponding lower limits
llims=replicate(0.,Npar)

;;= SET THE FITTED PARAMETERS
dustem_init_parinfo,pd,iv,up_limited=ulimed, $
      lo_limited=llimed,up_limits=ulims,lo_limits=llims

;;= Run the SED fitting
tol=1.e-14 ;Tolerance used
Nitermax=3 ;maximum number of iteration
      ;This is the criterium which will stop the fit procedure

loadct,13
;This is to ajust plot range
!y.range=[1e-4,10]
res=dustem_mpfite_sed(tol=tol,Nitermax=Nitermax)

;=== Plot best fit
yr=[1e-4,10] & xr=[1,2000]
tit='DUSTEM_Example'
ytit=textoidl('I_\nu_(M.Jy/sr)_for_N_H=10^{20}_H/cm^2')
xtit=textoidl('\lambda_(\mum)')
```

```

errors=(*(!dustem_fit).current_param_errors)*(*(!dustem_fit).param_init_values)
chi2=(*(!dustem_fit).chi2
rchi2=(*(!dustem_fit).rchi2

loadct,13
dustem_sed_plot,(*(!dustem_fit).current_param_values), $
                ytit=ytit,xtit=xtit,title=tit,yr=yr,xr=xr,/ysty,/xsty, $
                res=res,errors=errors,chi2=chi2,rchi2=rchi2

```

## 7 Routines Description

Most IDL routines return help by simply including the `/help` (or `help=1`) keyword in the call. An html description of the `DustemWrap` routines is available here:  
[http://www.cesr.fr/DUSTEM\\_WRAP/Routines.html](http://www.cesr.fr/DUSTEM_WRAP/Routines.html)

## 8 Dependencies

`DustemWrap` has dependencies towards the IDL Libraries listed below:

- Astron: <http://idlastro.gsfc.nasa.gov/>
- Cgis: <http://lambda.gsfc.nasa.gov/product/cobe/cgis.cfm>
- cmtotal: <http://www.physics.wisc.edu/~craigm/idl/idl.html>
- JPBLib: [http://www.cesr.fr/~bernard/JPB/Software/jpb\\_software.html](http://www.cesr.fr/~bernard/JPB/Software/jpb_software.html)
- Textoidl: <http://physweb.mnstate.edu/mcraig/textoidl/>

These libraries can be downloaded from the web. Alternatively, the routines used from these libraries are included in the `DustemWrap` package and can be added to the IDL path (see Sec. 1).

## 9 DustemWrap IDL system variables

The following lists all system variables used by the `DustemWrap` code.

- `!dustem_fit`: Includes SED fit parameters, such as variable and fixed parameters definition, initial values, etc ...
- `!dustem_data`: Contains SED data.
- `!dustem_inputs`: Used to tell `DustemWrap` which Fortran input file should be used as a default. Initialized in `dustem_init.pro` based on the `?` keyword. Used in `dustem_read_all_web3p8.pro`
- `!dustem_params`: Contains current parameters of the model.
- `!dustem_idl_continuum`: If set to 1, a NIR Black-Body continuum is added to the Fortran outputs.
- `!dustem_filters`: Contains filter information. This is initialized by `dustem_filter_init.pro`
- `!dustem_verbosity`: If set to 1, `DustemWrap` is verbose, otherwise quiet.
- `!dustem_show_plot`: If set to 1, plots are shown during SED minimization.
- `!dustem_inputs`

The following lists the content of non trivial system variables used by the DustemWrap code.

- !dustem\_fit:  
DATA : ?  
WAVELENGTH : ?  
PARAM\_DESCS : Variable parameter description  
PARAM\_INIT\_VALUES : Variable parameter initial value  
PARAM\_FUNC : ?  
FIXED\_PARAM\_DESCS: Fixed parameter description  
FIXED\_PARAM\_INIT\_VALUES: Fixed parameter value  
CHI2: Current  $\chi^2$   
RCHI2: Current reduced  $\chi^2$   
CURRENT\_PARAM\_VALUES: Current parameter value  
CURRENT\_PARAM\_ERRORS: Current parameter uncertainty value
  
- !dustem\_data: Contains SED data. INSTRU\_NAMES : Instrument name  
FILT\_NAMES : Filter names  
WAV : Wavelengths  
VALUES : Intensity values  
SIGMA :  $1 - \sigma$  uncertainty values
  
- !dustem\_inputs:
  
- !dustem\_params: Contains current parameters of the model. NGRAINS: number of grain types  
G0 : Value of the G0 parameter  
KEYWORDS : ?  
GRAINS : structures describing grain types  
ISRF : ISRF intensity array  
QABS : Qabs values for each grain type.  
CALOR : Heat capacities for each grain type  
LAMBDA : Wavelength used in emission calculations.  
SIZE : size distribution information for each grain type.  
MIX : Mixture information for each grain type.
  
- !dustem\_filters: Contains filter information. This is initialized by dustem\_filter\_init.pro  
IRAC : IRAC filters  
MIPS : MIPS filters  
MSX : MSX filters  
IRAS : IRAS filters  
DIRBE : DIRBE filters  
SPM : PRONAOS SPM filters  
ISOCAM : ISOCAM filters  
ISOPHOTP : ISOPHOTP filters  
ISOPHOTC : ISOPHOTC filters  
EFIRAS : Equivalent bandwidth for FIRAS used in ?  
ARCHEOPS : Archeops filters  
HFI : Planck HFI filters  
LFI : Planck LFI filters  
WMAP : WMAP filters  
SPIRE : Herschel SPIRE filters

PACS: Herschel PACS filters  
AKARI: AKARI filters

- !dustem\_inputs